
ABSTRACT
In this paper we illustrate a technical solution to the Interactive
Television Applications case study presented in [1] using languages
and tools from the Epsilon GMT component based on the infras-
tructure provided by the Eclipse Modeling Framework (EMF).

1. INTRODUCTION
The case study presented in [1] outlines the requirements of a sys-
tem that supports designing interactive television applications us-
ing model engineering techniques and principles. We have im-
plemented the main operations of such a system with languages
and tools from the Epsilon GMT component [2], using the Eclipse
Modeling Framework (EMF) [3] as the underlying modelling tech-
nology. The rest of the paper is organized as follows. In Section 2
we provide a brief discussion on Epsilon. In Section 3 we discuss
the implementation of model management operations involved in
the system, and in Section 4 we draw conclusions on the usefulness
of our approach and the needs for extending Epsilon, as identified
during the implementation process.

2. THE EPSILON MODEL MANAGEMENT
PLATFORM

Epsilon [2], is a platform of integrated task-specific model man-
agement languages, and a component of the Eclipse Generative
Modeling Technologies (GMT) research incubator project. Epsilon
provides languages for direct manipulation of models (EOL) [4],
model merging (EML) [5], model comparison (ECL) [6], model-
to-model transformation (ETL), model validation (EVL) [7] and
model-to-text transformation (EGL). All the languages of the plat-
form build on a common OCL-based model navigation and ma-
nipulation language (EOL) and a common runtime environment,
and are therefore highly interoperable. For example, anoperation
defined using EOL can be reused (imported) as-is by the model-to-
model and the model-to-text transformation languages. With regard
to supported modelling technologies, the architecture of Epsilon
allows users to manage models of different technologies, such as
MDR [8] and EMF [3] models and XML documents, and even fa-
cilitates support for additional model representation formats using

a layered architecture (discussed in [4]). In the context of tool-
support, Epsilon languages are supported by stable execution en-
gines and Eclipse-based development tools, available at [2].

3. IMPLEMENTING THE CASE STUDY
In this section we identify the model management operations in-
volved in the system and present a concrete implementation for
each. Our intention here is not to provide a ready-to-use tool, but
instead to demonstrate how the model-management-related com-
ponents of such a tool can be implemented using Epsilon. Inte-
grating the components into an application that can be operated by
end-users would require a significant amount of trivial Java code,
which is clearly out of scope of this work.

3.1 Designing the TVApp DSL
Our first step was to design a Domain Specific Language (DSL)
that enables users to design Interactive TV Applications. We have
designed our TVApp DSL atop EMF by defining its abstract syn-
tax in terms of ECore (using Emfatic [9] as a convenience textual
representation). A graphical overview of the DSL is presented in
Figure 1.

3.2 Editing Support for the TVApp DSL
The next step was to provide editing support for the DSL so that
users can compose models that conform to the language. For this
task the alternatives were: to use a tree-based editor; to use a tex-
tual syntax (e.g. with oAW XText [10]); and/or to use a diagram-
matical syntax (e.g. using GMF [11]). Since the TVApp DSL is
predominantly hierarchical, in the context of this case study we
decided to adopt the tree-based editor approach. EMF provides
a reflective tree-based editor that can be used to edit EMF mod-
els of arbitrary ECore metamodels. On top of this reflective ed-
itor we have implemented Exeed [12], an extension that enables
users to customize the appearance (e.g labels, icons, property view)
of the editor by attaching specificEAnnotation s toEClass es
andEStructuralFeature s of the metamodel. For instance,
to make both thename and theinformation of aText appear
on the editor (see Figure 3), the annotations displayed in Listing 1
were added to the definition of theText EClass .

Listing 1: Definition of the Text EClass using Emfatic
@exeed(classIcon="text",
label="return self.name + ’: ’ + self.information;")

class Text extends Content {

@exeed(multiLine="true")
attr String information;
val TextHistory[ * ] history;

}



Figure 1: The TVApps Domain Specific Language

Exeed provides a number of additional annotations discussed in
detail in [12]. In fact, the same customization can be achieved by
generating a dedicated editor for the DSL and then customizing it
using Java. However, as discussed in [12], we regard using Ex-
eed as a more agile approach, particularly for developing prototype
editors. The basic and enhanced views of a TVApp model in the re-
flective editor and Exeed are shown in Figures 2 and 3 respectively.

3.3 Generating a TVApp model for a Sports
Competition

To address Case 1 of [1], we have designed a new Competition
metamodel for capturing information about competitions, groups
and participating teams. A graphical overview of the metamodel is
displayed in Figure 4. The design of the metamodel also satisfies
the requirement of Case 2 for defining the team-related informa-
tion first and then adding the teams into groups by modifying the
competitors list of each group.

To transform a Competition model into a TVApp model, we have
used the Epsilon Transformation Language (ETL) [2]. The trans-
formation displayed in Listing 2 defines that anApplication
and aVote will be generated from eachCompetition , and that
the Vote will contain a flattened sequence ofLabel s (one for
eachGroup ) and Choice s (one for eachCompetitor in the
group).

Listing 2: ETL transformation that transforms a Competition
model into a TVApp model
rule Competition2Application

transform c : Competition!Competition
to a : TVApp!Application, v : TVApp!Vote {

a.name := c.name + ’ Application’;
v.name := ’Who will win the ’ + c.name + ’?’;

Figure 2: Designing TVApp models using the Reflective Editor

Figure 3: Designing TVApp models using Exeed



Figure 4: The Competition Domain Specific Language

a.contents.add(v);

for (g in c.groups) {
v.contents.add(g.equivalent());
for (memb in g.members) {

v.contents.add(memb.equivalent());
}

}
}

rule Competitor2Choice
transform co : Competition!Competitor
to ch : TVApp!Choice {

ch.name := co.name;
}

rule Group2Label
transform g : Competition!Group
to l : TVApp!Label {

l.name := ’Group ’ + g.name;
}

3.4 Integrating Live Reports
Case 3 of [1] requires that an external user must be able to update
a particular text but not the structure of the application. To achieve
this we have defined the minimal Report DSL. The rationale is that
a user can be granted the permission to only compose Report mod-
els which will then be merged with the original application model.
To satisfy the requirement set in Case 4 of [1], when merging the
TVApp with a Report, the original information contained in the
Text s which the Report updates are not lost but instead are stored
in the form of newTextHistory model elements. We imple-
ment the merging using the Epsilon Merging Language (EML) [5]
module displayed in Listing 3.

As discussed in [5], EML operates in two steps. The first step is to
establish correspondences between elements that need to be merged
(using match-rules) and the second is to merge those elements (us-
ing merge-rules) and optionally transform any non-matched ele-
ments into the target model (using transform-rules reused from ETL).
In Listing 3, matchingReport s from the Report model withText s
from an OldTVApp model is achieved through theMatchReport

Figure 5: The Report Domain Specific Language

WithText match-rule that compares the names of the compared
elements. Then, matchingReport s andText s are merged using
theMergeReportWithText merge-rule which specifies that the
Text should be updated and that the old version of the text should
be stored in the form of a newTextHistory model element with
a proper information and revision number.

Listing 3: EML module that merges a TVApp with a Report
model
rule MatchReportWithText

match t : OldTVApp!Text
with r : Report!Report {

compare : r.name = t.name
}

auto rule MergeReportWithText
merge ot : OldTVApp!Text
with r : Report!Report
into nt : NewTVApp!Text {

var h : new NewTVApp!TextHistory;
h.information := ot.information;
h.revision := ot.history.collect(h|h.revision).max(0) + 1;
nt.history.add(h);
nt.information := r.information;

}

3.5 Generating XML from a TVApp model
To generate XML from a TVApp model, as required in Section 6 of
[1], we use the pivot XML metamodel displayed in Figure 6. Thus,
we transform the TVApp model into an XML model (Listing 4)
using the Epsilon Transformation Language, and then we generate
the textual representation of the XML model (Listing 5) using the
Epsilon Generation Language, a template-based language for text
generation. This approach promotes modularity and reusability by
targeting XML-specific issues such as escaping and indentation in
a separate transformation which can be reused as-is in other DSL
→ XML scenarios, and also complies with the requirement not to
use a special (hard-coded) method for transforming XML models
to text.

Listing 4: ETL transformation that transforms TVApp models
to XML models
pre {

var doc : new Xml!Document;
doc.rootElement := TVApp!Application.

allInstances.first().equivalent();
}

abstract rule NamedElement2Element
transform ne : TVApp!NamedElement
to n : Xml!Element {

n.addAttribute(’name’, ne.name);
}

rule Application2Element
transform a : TVApp!Application



Figure 6: The XML Domain Specific Language

to n : Xml!Element extends NamedElement2Element{

n.name := ’Application’;
n.contents := a.contents.equivalent();

}

rule Vote2Element
transform v : TVApp!Vote
to n : Xml!Element extends NamedElement2Element {

n.name := ’Vote’;
n.contents := v.contents.equivalent();

}

rule Choice2Element
transform c : TVApp!Choice
to n : Xml!Element extends NamedElement2Element {

n.name := ’Choice’;
}

rule Label2Element
transform c : TVApp!Label
to n : Xml!Element extends NamedElement2Element {

n.name := ’Label’;
}

rule Text2Element
transform t : TVApp!Text
to e : Xml!Element extends NamedElement2Element {

e.name := ’Text’;
var text : new Xml!Text;
text.cdata := t.information;
e.contents.add(text);

}

rule Menu2Element
transform m : TVApp!Menu
to e : Xml!Element extends NamedElement2Element {

e.name := ’Menu’;
e.contents := m.contents.equivalent();

}

operation Xml!Element addAttribute
(name : String, value : String) {
var attr : new Xml!Attribute;
attr.name := name;
attr.value := value;
self .attributes.add(attr);

}

The ETL transformation of Listing 4, demonstrates two important
characteristics of ETL: rule inheritance and state-changing opera-
tions. By inheriting from the abstract ruleNamedElement2Element,
the rest of the rules are maintained simple and without duplica-
tion. Moreover, defining theaddAttribute()state-changing opera-
tion simplifies theNamedElement2Elementrule.

Listing 5: EGL template that generates XML text from XML
models
<?xml version="1.0"?>
<! --Generated using EGL-->
[%=Document.allInstances().first().rootElement.toString(0)%]

[%
operation Element toString(indent : Integer) : String {

var str : String;
str := indent.getIndent() + ’<’ + self .name.normalize();
for (a in self .attributes) {

str := str + ’ ’ + a.name.normalize() +
’="’ + a.value.normalize() + ’"’;

if (hasMore){
str := str + ’ ’;

}
}
str := str + ’>\r\n’;
for (c in self .contents) {

str := str + c.toString(indent + 1);
}
str := str + indent.getIndent() + ’</’ +

self .name.normalize() + ’>\r\n’;
return str;

}

operation Text toString(indent : Integer) : String {
return (indent + 1).getIndent() +

self .cdata.normalize() + ’\r\n’;
}

operation Integer getIndent() : String {
var indent : String;
for (i in Sequence{1 .. self }){

indent := indent + ’ ’;
}
return indent;

}

operation String normalize() {
var normalized : String := self ;
if ( not normalized.isDefined()) {

normalized := ’’;
}
else {

normalized := normalized.replace(’&’, ’&amp;’);
normalized := normalized.replace(’<’, ’&lt;’);
normalized := normalized.replace(’>’, ’&gt;’);
-- etc

}
return normalized;

}
%]

Generating XML text from an XML model in Listing 5 involves
much dynamic and little static text. Therefore, the vast majority of
the serialization process is performed via string concatenation. By
contrast, the header (processing instruction and comment) is only
static text that is emitted as-is from the template.

3.5.1 Integrating the Model-to-Model and the Model-
to-Text steps

To integrate the two steps presented above into a coherent process
that transforms a TVApp model directly to textual XML, we use the



ANT-based Epsilon Workflow [13]. Thus, in Listing 6, we define
two tasks for loading the involved models (epsilon.loadModel)1,
one that invokes the ETL transformation (epsilon.etl) and one that
invokes the EGL model-to-text transformation on the intermediate
XML model (epsilon.egl).

Listing 6: The wokflow that integrates the ETL and EGL tasks
1 <?xml version ="1.0"?>
2 <project default ="main">
3
4 <epsilon.loadModel name="TVApp" type="EMF">
5 <parameter name="modelFile"
6 value="models/ChampionsLeague.ecore"/>
7 <parameter name="metamodelFile"
8 value="../metamodels/TVAppDsl.ecore"/>
9 <parameter name="isMetamodel" value="false"/>

10 <parameter name="isMetamodelFileBased" value="true"/>
11 <parameter name="readOnLoad" value="true"/>
12 </epsilon.loadModel>
13
14 <epsilon.loadModel name="Xml" type="EMF">
15 <parameter name="modelFile"
16 value="models/TVAppXml.ecore"/>
17 <parameter name="metamodelFile"
18 value="../metamodels/Xml.ecore"/>
19 <parameter name="isMetamodel" value="false"/>
20 <parameter name="isMetamodelFileBased" value="true"/>
21 <parameter name="readOnLoad" value="false"/>
22 </epsilon.loadModel>
23
24 <target name="main">
25
26 <epsilon.etl src="TVApp2Xml.etl" sourcemodel="TVApp"
27 targetmodel="Xml">
28
29 <model ref="TVApp"/>
30 <model ref="Xml"/>
31 </epsilon.etl>
32
33 <epsilon.egl src="Xml2Text.egl"
34 target="output/TVApp.xml">
35 <model ref="Xml"/>
36 </epsilon.egl>
37
38 </target>
39
40 </project>

3.6 Generating a Mock-up of the TV Applica-
tion

A feature that is not required explicitly in [1], but which we found
particularly useful is to be able topreviewa TVApp model using
a mock-up that closely resembles the appearance of the final de-
ployed application. Therefore, we have used EGL to compose a
model-to-text transformation that generates a set of linked HTML
screens that emulate the look-and-feel and functionality of the de-
ployed application2. Listing 7 presents the main template of the
EGL solution that iterates the model and invokes the respective sub-
templates (Text.egl, Menu.egl, Vote.egl) to generate the mockup
HTML screens. The complete source code for this and all the
other model management tasks presented in this paper is available
at http://www.eclipse.org/gmt/epsilon/doc.

1An alternative to hard-coding the location where the TVApp
model resides (line 6 of Listing 6) would be to use an external ANT
parameter that users can set from the invocation environment (e.g.
Eclipse launch configuration, command line)
2An exemplar mockup generated from the TVApp
model displayed in Figure 3 is available at
http://www.cs.york.ac.uk/∼dkolovos/epsilon/mdd-
tif/Champions%20League.html

Listing 7: EGL template that generates mockup HTML screens
for a TVApp model
[%

import ’include\\Common.eol’;

TemplateFactory.setTemplateRoot(’workspace\\MDD-TIF’);
TemplateFactory.setOutputRoot(’workspace\\MDD-TIF\\html’);

var header : Template :=
TemplateFactory.load(’include\\Header.egl’);

var text : Template :=
TemplateFactory.load(’include\\Text.egl’);

var menu : Template :=
TemplateFactory.load(’include\\Menu.egl’);

var vote : Template :=
TemplateFactory.load(’include\\Vote.egl’);

var footer : Template :=
TemplateFactory.load(’include\\Footer.egl’);

var contents : String := ’default’;

if (Application.isType(content) or Menu.isType(content)) {
menu.populate(’menu’, content);
contents := menu.process();

-- Recursively generate the pages
-- for the contents of this
-- application / menu
for (child in content.contents) {

var page : Template := TemplateFactory.load(’Page.egl’);

page.populate(’content’, child);

page.generate(child.filename());
}

} else {
if (Text.isType(content)) {

text.populate(’text’, content);
contents := text.process();

} else {
if (Vote.isType(content)) {

vote.populate(’vote’, content);
contents := vote.process();

}
}

}

%]
[%=header.process()%]
[%=contents%]
[%=footer.process()%]

4. CONCLUSIONS
In this paper we have demonstrated using languages and tools from
the Epsilon GMT component to implement the case study presented
in [1].

Summarizing our experiences from the implementation process, we
have found implementing the model management tasks using lan-
guages of the Epsilon platform to be quite straightforward. How-
ever, by implementing the case study we also realized that while it
is easy to implement and execute model management tasks from a
developer perspective, to make it accessible to end-users, a signif-
icant amount of Java code must also be written - particularly with
respect to presenting the available tasks to the users and enabling
them to invoke the tasks in a user-friendly manner. In our view, this
is a very important aspect that must be addressed in a generic way
to contribute to the wide-spread use of model engineering.

